
1 

Paper 149-31 

An Alternative Method of Producing Labels without the FORMS 
Procedure  

Dongming Jiang, University of Cincinnati, Cincinnati, OH 
 

ABSTRACT  
Many users of SAS® System software, especially those who want to print labels for mailing labels or external tape 
labels, are often confronted with difficulties on size and spacing. How to perfectly print names and addresses on 
standard labels such as those produced by Avery®? 
 
While the PROC FORMS procedure is the traditional procedure to generate a variety of labels, it may not be 
possible to successfully obtain the desired output which is accurately suit the standard Avery® label product.  
 
The techniques presented in the paper provide the user with more control on the placement parameters and fonts 
to facilitate printing data sets to labels. The features of the two methods are compared and discussed. 
 
The ideas presented here were developed and tested using SAS® Release 8.2 (TS2M0) running under Windows 
XP®. 

INTRODUCTION 
Some users of SAS System software l ike maintaining an address book in a SAS data set. To facilitate printing data 
sets to labels, there is a need to obtain the desired output which fits a specific format for the standard labels such 
as those produced by Avery®. This can be achieved through a series of PROC TEMPLATE, DATA STEPS and 
PROC REPORT. Proc Template allows the user to redefine many elements of a custom document. Proc Report 
provides a wide array of formatting capabilities within its own domain. 

SAMPLE DATA SET 
The sample data set contains the names (first name, last name) and mailing addresses. The objective is to print 
one set of mailing labels consisting of one copy of the form unit for each observation on Avery® 11416....20 down, 
4 across. 
 

data list; 
   input FirstName $ 1-13 LastName $15-22 University $ 24-45; 
   datalines; 
Jennifer      Cappel   Unive. of  Missouri    
Marc          Berman   Unive. of  Cincinnati  
Adam          Bell     Unive. of  Kentucky    
Denise        Jack     Unive. of  Washington  
James         Jordan   Unive. of  Michigan    
Joe           Kaiser   Unive. of  Wisconsin   
Stacy         Lampe    Unive. of  Georgia     
Amy           Marshall Unive. of  Chicago     
Mark          Martin   Unive. of  Arizona     
Grace         Rice     Unive. of  California  
; 
run; 

PROC FORMS 
The PROC FORMS procedure could be used to generate a variety of labels 
 

filename labels "C:\Forms.txt"; 

      proc forms data=List file=labels 
           ndown=20 
           align=0 
           width=22 
           across=4 
           down=0 
           skip=1 
         between=2 
           ; 
           line 1 FirstName LastName; 
           line 2 University; 
      run; 

 
The FILENAME statement associates the name LABELS with the external file that will receive the output from 
PROC FORMS.  
 
FILE:             sends the output to the file associated with the fileref LABELS.  
NDOWN:       prints 20 rows of form units on each page.  
ALIGN:          suppresses the printing of dummy form units.  
WIDTH:         sets the width of the form units to 22. 
ACROSS:      writes 4 form units across each page. 
DOWN:         suppresses the lines at the top of the file. 
SKIP:             skips 1 line between form units to maintain the proper alignment. 
BETWEEN:   puts 2 blank characters between adjacent form units. 
The LINE statements specify the variables to place on each line. First Name and Last Name would be shown on 
the fist line. Universities are on the second line; 
 
The size and spacing of form units are controlled by options such as DOWN, INDENT, WIDTH, LINES, BETWEEN 
as shown in below 

 
Since these options were integers (not inches) and were used to specify the number of spaces or the number of 
lines. For some cases, it may not be possible to successfully make the perfect margin and size which suit the 
standard Avery® label product. Although some people have even thought about putting the data set out to a flat file 
and then reading it with Word but this too seems tedious. 
 

ALTERNATIVE METHOD TO PROC FORMS 
The alternative method could be summarized as the following series of steps: 
1. Fixing the margins. 

2. Create 4 column variables for the 4 form units across each page. 
3. Create 3 column variables with blank characters to separate the above 4 form units.  
4. Use PROC REPORT output the dataset to the RTF destination.  
 
1. Fixing the margins. To perfectly print labels on Avery® 11416, the first thing to modify is the margins. To set 
margins PROC TEMPLATE creates a style to which the ODS statement refers. In the following code, we use our 
custom template to get 0.25-inch margins at the top and bottom, 0.25-inch margins on the right side, and 0.3-inch 
margins on the left side. 
 

proc template;                                                                 
   define style newstyle;                                                    
      parent = styles.printer;                                                 
      replace Body from Document                                               
         "Controls the Body file." /                                           
         bottommargin = 0.25in                                                 
         topmargin = 0.25in                                                    
         rightmargin = 0.25in                                                  
         leftmargin = 0.3in;                                                  
   end;                                                                        
run;        
 

2. Create 4 column variables for the 4 form units across each page. 
 

Data PLabel1; Set list; 
Row1=Trim(FirstName) || '    ' || Trim(LastName); 
Row2=Trim(University); 
keep Row1 Row2; 
run; 
 
Data PLabel2; Set PLabel1; 
  Cell=Trim(Row1) || '                                 ' || Trim(Row2); 
run; 
 
Data PLabel3; Set PLabel2; 
CIndex=mod(_N_,4); 
keep CIndex Cell; 
run; 
 
Data PLabelC1 PLabelC2 PLabelC3 PLabelC4; Set PLabel3; 
     if CIndex=1 then output PLabelC1; 
     if CIndex=2 then output PLabelC2; 
     if CIndex=3 then output PLabelC3; 
     if CIndex=0 then output PLabelC4; 
run; 
data PLabelC1;set PLabelC1; 
rename Cell=Tab1; 
IND=_N_; 
run; 
data PLabelC2;set PLabelC2; 
rename Cell=Tab2; 
IND=_N_; 
run; 
data PLabelC3;set PLabelC3; 
rename Cell=Tab3; 

IND=_N_; 
run; 
data PLabelC4;set PLabelC4; 
rename Cell=Tab4; 
IND=_N_; 
run; 
proc sort data=PLabelC1; by IND; 
proc sort data=PLabelC2; by IND; 
proc sort data=PLabelC3; by IND; 
proc sort data=PLabelC4; by IND; 
run; 
data PLabel4; merge PLabelC1 PLabelC2 PLabelC3 PLabelC4;by IND; 
run; 

 
 
3. Create 3 column variables with blank characters to separate the above 4 form units. Here the goal is to put 
some blank characters between adjacent form units. PLabel5 is the final data set that would be used for printing. 
 

Data Plabel5; Set Plabel4; 
   Tab1b="-----"; 
   Tab2b="-----"; 
   Tab3b="-----"; 
run;  
 
proc format; 
  value $between '-----' = 'A0'x'A0'x'A0'x'A0'x'A0'x'A0'x 
        ; 
run; 
data PLabel5; set Plabel5; 
format Tab1b Tab2b Tab3b $between. ; 
run; 
 

4. Use PROC REPORT output the dataset to the RTF destination. 
 

title; *remove the default SAS title line; 
 
ODS RTF FILE="C:\Label.doc" bodytitle nokeepn style=newstyle;  
ODS NOPTITLE;  
 
proc report data=PLabel5 nowd headline split='?' ps=40 NOHEADER  
  style=[FRAME=VOID RULES=NONE] 
  style(column)=[CELLHEIGHT=0.492in CELLPADDING=15%] ; 
  column  Tab1 Tab1b Tab2 Tab2b Tab3 Tab3b Tab4 ; 
  define Tab1 / left  width=20 FLOW  'TAB 1'; 
  define Tab1b / left  width=5 FLOW  '-----'; 
  define Tab2 / left  width=20 FLOW SPACING=1 'TAB 2'; 
  define Tab2b / left  width=5 FLOW  '-----'; 
  define Tab3 / left  width=20 FLOW SPACING=1 'TAB 3'; 
  define Tab3b / left  width=5 FLOW  '-----'; 
  define Tab4 / left  width=20 FLOW SPACING=1 'TAB 4'; 
run; 
Ods rtf close; 

The TITLE statement removes the default SAS title, "The SAS System". In the domain of Proc Report, FRAME=VOID 
removes the borders. The CELLHEIGHT option specifies the height of each cell, i.e. the height of each form unit. 
The CELLPADDING option specifies the space between table cell contents and the cell border. The Width option 
defines the width of the column. The FLOW option wraps character variables onto multiple lines so that in this 
case, First Name together with Last Name would be shown on the fist line and Universities are on the second line; 
 
 
Final Output: 

 
 

ADVANTAGES 
Several advantages using the alternative method were found.  
• It allows for greater control on the size and spacing parameters.  
• With suitable settings, it could be used repeatedly for specific Avery® label product. 
• It outputs the labels to the Word file directly. In addition, users don’t need to modify anything in that Word file. 
 

DISADVANTAGES 
• Compared with the FORMS procedure. The proposed method requires more work on the DATA STEP. 
 

CONCLUSION  
Based on the PROC REPORT, PROC TEMPLATE and DATA STEPS, this paper presents the techniques which 
provide the user with more control on the placement parameters and fonts to facilitate printing data sets to labels. 

 

CONTACT INFORMATION  
Your comments and questions are valued and encouraged.  Contact the author at: 
 

Dongming Jiang 
OLD CHEM BLDG,  
Department of Mathematical Sciences 
University of Cincinnati  
Cincinnati, OH 45221-0025 
Email: jiangd@math.uc.edu 

